Agilent AN 369-5
Multi-frequency C-V Measurement of Semiconductors

Application Note

Agilent E4980A and 4284A Precision LCR Meters

Introduction

Parameters such as the capacitance of the oxide layer (Cox) and the density of substrate impurities (Nsub) that are required in the evaluation of the manufacturing process of MOS type semiconductors can be derived by using measured C-V characteristics. To make an accurate evaluation of these processes, precise C-V measurements are required. Such measurements entail the following difficulties.

Difficulties

1. There is no single instrument that can make C-V measurements from low to high frequencies.
2. It is difficult to compensate for the additional errors that occur when cable extensions or a prober are used.
3. The accuracy and reliability of the DC bias voltage are not sufficient to perform repeatable C-V measurements.

In the following discussion we will show how C-V measurements are performed with the E4980A and 4284A solving these problems.

Solutions Offered by the E4980A and 4284A

1. Wide Frequency Range Measurements from 20 Hz to 2 MHz.

The program listing shown in the appendix was used to measure the C-V characteristics at 10 kHz, 100 kHz, and 1 MHz of the MOS device whose characteristics are shown in Figure 1.

Thus, the E4980A and the 4284A can single-handedy perform C-V measurements in both the low and high frequency ranges. This allows it to measure high-loss devices (semiconductors on large diameter wafers, etc.), which are difficult to measure at 1 MHz, at low frequencies (10 kHz, 100 kHz, etc.).



Figure 1. Multi-frequency C-V characteristics of a semiconductor

2. Compensation Functions for Cable Extensions and Probers.

In order to test semiconductor devices on a wafer, an extension cable and a prober are required. (See Figure 2.) The extension cable and the prober cause additional errors that greatly influence the test value. The E4980A and 4284A's 2m/4m Cable Length Operation function with 16048D/E test leads, Option 006 for 4284A, minimizes additional errors caused by using 2m/4m extension cables. The E4980A and 4284A's open/short/load compensation functions minimize errors caused by a prober. This powerful compensation function ensures highly accurate C-V measurements even when a prober is used.



Figure 2. Extension cable and prober

3. Highly Accurate Internal Bias

Insufficient accuracy and stability of the bias voltage applied to a device prevents accurate C-V measurements. The E4980A and 4284A ensure the application of a stable bias voltage with a maximum accuracy of 0.1% (Option 001). This minimizes measurement errors due to bias voltage errors.

Conclusion

The E4980A and 4284A feature a wide frequency range, powerful compensation functions, and highly accurate bias characteristics. This versatility of the meter guarantees highly accurate C-V measurements, and higher efficiency in the evaluation of semiconductor processes. Because the E4980A and 4284A can perform all of the low and high frequency C-V measurements singlehandedly, it will greatly contribute to the lowering of capital costs.

Table 1. E4980A Option 001 DC Bias Capability

Range	Resolution	Accuracy*
±0 to 5 V	330 µV	± 0.1% + 2 mV
±5.001 to 10 V	1 mV	± 0.1% + 2 mV
±10.002 to 20 V	2 mV	± 0.1% + 2 mV
±20.005 to 40 V	5 mV	± 0.1% + 2 mV

*This can only be used when the test signal level is < 2 Vrms

Table 2. E4984A Option 001 DC Bias Capability

Range	Resolution	Accuracy*
±0.000 to 5 V	1 mV	± 0.1% + 1 mV
±4.002 to 8 V	2 mV	± 0.1% + 2 mV
±8.005 to 20 V	5 mV	± 0.1% + 5 mV
±20.01 to 40 V	10 mV	± 0.1% + 10 mV

*This can only be used when the test signal level is < 2 Vrms

Appendix. Sample Program

```
10   ! ******* Agilent 4284A C-V MEASUREMENT SAMPLE PROGRAM *******
20   !
30   GCLEAR
40   PRINT CHR$(12)                                ! CLEAR DISPLAY
50   DIM C(3,128),Cn(3,128),Vbias(128)
60   REAL Datab(1:513) BUFFER
70   DIM Dum$[63] BUFFER
80   ASSIGN @Ad TO 717;FORMAT ON                   ! DEFINE I/0 PATH
90   ASSIGN @Pat TO 717;FORMAT OFF                 ! DEFINE I/0 PATH
100  ASSIGN @Dummy TO BUFFER Dum$                  ! DEFINE BUFFER FOR HEADER
110  ASSIGN @Buff TO BUFFER Datab(*)               ! DEFINE BUFFER FOR DATA
120  REMOTE @Ad                                    ! SET 4284A TO REMOTE MODE
130  OUTPUT @Ad;"*RST"                             ! RESET 4284A
140  OUTPUT @Ad;"VOLT:LEV 0.3V"                    ! SET OSC LEVEL TO 300mV
150  OUTPUT @Ad;"OUTP:HPOW ON"                     ! BIAS OPTION ON
160  OUTPUT @Ad;"TRIG:SOUR BUS"                    ! GPIB TRIGGER MODE
170  OUTPUT @Ad;"FORM:DATA REAL"                   ! BINARY DATA TRANSFER
160  OUTPUT @Ad;"MEM:DIM DBUF,101"                 ! DEFINE BUFFER IN 4284A
190  OUTPUT @Ad;"MEM:FILL DBUF"                    ! ENABLE BUFFER TRANSFER
200  OUTPUT @Ad;"APER SHOR"                        ! SET INTEG TIME TO SHORT
210  GOSUB Corr
220  PRINT "CONNECT OUT AND PRESS CONTINUE"
230  PAUSE
240  PRINT CHR$(12)                                ! CLEAR DISPLAY
250  Holdtime=10                                   ! HOLD TIME = 10sec
260  Delaytime=.1                                  ! DELAY TIME = 100msec
770  Vbias(1)=-5                                   ! START BIAS VOLTAGE
280  Vstep=.1                                      ! 100mV BIAS STEP
290  OUTPUT @Ad;"TRIG:DEL "&VAL$(Delaytime)        ! SET DELAY TIME
300  OUTPUT @Ad;"BIAS:STATE ON"                    ! BIAS ON
310  Freq$(1)=" 1MHz"
320  Freq$(2)=" 100KHz"
330  Freq$(3)=" 10KHz"
340  FOR I=1 TO 3
350    OUTPUT @Ad;"FREQ"&Freq$(I)                  ! FREQUENCY SETTING
360    OUTPUT @Ad;"BIAS:VOLT "&VAL$(Vbias(1))     ! SET START BIAS VOLTAGE
370    PRINT TABXY(5,15);"HOLD TIME 10SEC"
380    WAIT Holdtime                               ! HOLD TIME
390    PRINT TABXY(5,15);"MEASURING at "&Freq$(I)
400    FOR J=1 TO 101
410      OUTPUT @Ad;"TRIG"                         ! TRIGGER 4284A
420      IF J=101 THEN 450
430      Vbias(J+1)=Vbias(J)+Vstep                 ! CHANGE BIAS VOLTAGE
440      OUTPUT @Ad;"BIAS:VOLT "&VAL$(Vbias(J+1))  ! SET BIAS VOLTAGE
450    NEXT J
460    RESET @Dummy
470    RESET @Buff
480    OUTPUT @Ad;"MEM:READ? DBUF"                 ! READ BUFFER
490    TRANSFER @Pat TO @Dummy;COUNT 6,WAIT        ! TRANSFER HEADER
500    TRANSFER @Pat TO @Buff;COUNT 101*4*8+1 ,WAIT ! TRANSFER DATA
510    FOR J=1 TO 101
520      C(I,J)=Datab(J*4-3)
530    NEXT J
540    IF MAX(C(*))=0 THEN Err                     ! CHECK IF MAX C VALUE IS 0
550    FOR J=1 TO 101
560      Cn(I,J)=C(I,J)/MAX(C(*))                  ! NORMALIZE C VALUE
570  ! PRINT Cn(I,J) ! PRINT NORMALIZED C VALUE
580    NEXT J
590    OUTPUT @Ad;"MEM:CLE DBUF;FILL DBUF"         ! CLEAR DATA IN BUFFER
600    PRINT CHR$(12)                              ! CLEAR DISPLAY
610  NEXT I
620  OUTPUT @Ad;"BIAS:STATE OFF"                   ! BIAS OFF
630  GOSUB Plotting
640  GOTO Endinq
650  !
66O  ! ********************** PLOT C-V **********************
670  ! THE FOLLOWINGS ARE FOR PLOTTING THE MEASUREMENT DATA TO CRT.
680  !
690  Plotting:DEG                                  ! SET ANGLE UNIT TO DEGREE
700  GRAPHICS ON
710  VIEWPORT 8,150,17,100                         ! PLOT NUMBERS AND LABELS
720  WINDOW 0,100,0,100                            !
730  LDIR 0                                        !
740  LORG 2                                        !
7SO  CSIZE 4                                       !
760  MOVE 50,15                                    !
770  LABEL "BIAS (V)"                              !
780  MOVE 8,23                                     !
790  LABEL "-5.0"                                  !
800  MOVE 54,23                                    !
810  LABEL "0"                                     !
820  MOVE 94,23                                    !
830  LABEL "5.0"                                   !
840  MOVE 5,28
```

```
850  LABEL "0.0"                                !
860  MOVE 5,98                                  !
870  LABEL "1.0"                                !
880  MOVE 75,5                                  !
890  LABEL " 1MHz"                              !
900  MOVE 75,10                                 !
910  LABEL "100kHz"                             !
920  MOVE 75,15                                 !
930  LABEL " 10kHz"                             !
940  MOVE 88,5                                  !
950  DRAW 98 ,5                                 !
960  LINE TYPE 8                                !
970  MOVE 88 ,10                                !
980  DRAW 98 ,10                                !
990  LINE TYPE 3                                !
1000 MOVE 88,15                                 !
1010 DRAW 98,15                                 !
1020 LINE TYPE 1                                !
1030 LDIR 90                                    !
1040 MOVE 3,55                                  !
1050 LABEL "Cap/Cox"                            !
1060 LDIR 0                                     !
1070 MOVE 5,28                                  !
1080 LABEL "0.0"                                !
1090 MOVE 5,98                                  !
1100 LABEL "1.0"                                ! END PLOT NUMBERS AND LABELS
1110 !
1120 VIEWPORT W3,125,40,100                     ! DRAW AXES
1130 FRAME                                              !
1140 WINDOW -5,5,0,1                            !
1150 AXES 1,.2,-5,0                             ! END DRAW AXES
1160 !
1170 FOR I=1 TO 3                               ! PLOT DATA
1180 MOVE Vbias(1),Cn(I,1)                      !
1190 FOR J=2 TO 101                             !
1200 DRAW Vbias(J),Cn(I,J)                      !
1210 NEXT J                                     !
1320 LINE TYPE 3                                !
1230 IF I=1 THEN LINE TYPE 8                    !
1240 NEXT I                                     ! END PLOT DATA
1250 LINE TYPE 1
1260 RETURN
1270 !
1280 Corr:! ************* SUBROUTINE FOR CORRECTION *************
1290 ! THE FOLLOWINGS ARE FOR PERFORMING OPEN/SHORT COMPENSATION.
1300 !
1310 OUTPUT @Ad;"DISP:PAGE CSETUP"              ! GO TO CORRECTION SETUP PAGE
1320 OUTPUT @Ad;"CORR:LENG O;METH SING"         ! CABLE LENGTH Om, SINGLE COMPEN MODE
1330 OUTPUT @Ad;"CORR:SPOT1:FREQ 1MAHz;STAT ON" ! SPOT FREQ 1 = 1MHz
1340 OUTPUT @Ad;"CORR:SPOT2:FREQ 100KHz;STAT ON" ! SPOT FREQ 2 = 100kHz
1350 OUTPUT @Ad;"CORR:SPOT3:FREQ 10KHz;STAT ON" ! SPOT FREQ 3 = 10kHz
1360 OUTPUT @Ad;"CORR:LOAD:STAT OFF"            ! LOAD COMPEN OFF
1370 PRINT "OPEN COMPENSATION"
1380 PRINT TABXY(5,15);"PRESS CONTINUE"
1390 PAUSE
1400 PRINT TABXY(5,15);" "
1410 OUTPUT @Ad;"CORR:SPOT1:OPEN"               ! PERFORM OPEN COMPEN AT SPOT FREQ 1
1420 OUTPUT @Ad;"CORR:SPOT2:OPEN"               ! PERFORM OPEN COMPEN AT SPOT FREQ 2
1430 OUTPUT @Ad;"CORR:SPOT3:OPEN"               ! PERFORM OPEN COMPEN AT SPOT FREQ 3
1440 OUTPUT @Ad;"*OPC?"                         ! CONFIRM OPEN MEASUREMENT COMPLETED
1450 ENTER @Ad;A$
1460 OUTPUT @Ad;"CORR:OPEN:STAT ON"             ! OPEN COMPEN ON
1470 PRINT CHR$(12)                             ! CLEAR DISPLAY
1480 PRINT "SHORT COMPENSATION"
1490 PRINT TABXY(5,15);"PRESS CONTINUE"
1500 PAUSE
1510 PRINT TABXY(5,15);" "
1520 OUTPUT @Ad;"CORR:SPOT1:SHOR"               ! PERFORM SHORT COMPEN AT SPOT FREQ 1
1530 OUTPUT @Ad;"CORR:SPOT2:SHOR"               ! PERFORM SHORT COMPEN AT SPOT FREQ 2
1540 OUTPUT @Ad;"CORR:SPOT3:SHOR"               ! PERFORM SHORT COMPEN AT SPOT FREQ 3
1550 OUTPUT @Ad;"*OPC?"                         ! CONFIRM SHORT MEASUREMENT COMPLETED
1560 ENTER @Ad;A$
1570 OUTPUT @Ad;"CORR:SHOR:STAT ON')            ! SHORT COMPEN ON
1580 OUTPUT @Ad;"DISP:PAGE MEAS"                ! GO TO MEASUREMENT PAGE
1590 PRINT CHR$(12)                             ! CLEAR DISPLAY
1600 RETURN
1610 !
1620 ! ********************************************************
1630 Err:PRINT "C-V MEASUREMENT WAS FAILED."
```

www.agilent.com

Remove all doubt

Our repair and calibration services will get your equipment back to you, performing like new, when promised. You will get full value out of your Agilent equipment throughout its lifetime. Your equipment will be serviced by Agilent-trained technicians using the latest factory calibration procedures, automated repair diagnostics and genuine parts. You will always have the utmost confidence in your measurements.

Agilent offers a wide range of additional expert test and measurement services for your equipment, including initial start-up assistance onsite education and training, as well as design, system integration, and project management.

For more information on repair and calibration services, go to

www.agilent.com/find/removealldoubt

Agilent Email Updates

www.agilent.com/find/emailupdates
Get the latest information on the products and applications you select.

Agilent Direct

www.agilent.com/find/quick
Quickly choose and use your test equipment solutions with confidence.

Product specifications and descriptions in this document subject to change without notice.

Printed in USA, October 2, 2006
5950-2953

Agilent Technologies